

## Potential Inside and Outside Sphere with Surface Potential $V[\theta] = V_0 \sin[\theta]^4$

### PROBLEM:


Consider a hollow, nonconducting sphere of radius  $R$  with a surface potential given by:

$$V[\theta] = V_0 \sin[\theta]^4 .$$

Find the electric potential and the electric field  $V[r, \theta]$  both inside and outside the sphere.

Execute the following code to view a surface-density plot of this potential.

```
In[6]:= ClearAll["Global`*"] (*Leave code for user to execute*)
ParametricPlot3D[{Sin[θ] Cos[φ], Sin[θ] Sin[φ], Cos[θ]}, {θ, 0, π}, {φ, 0, 2 π},
  ColorFunction → Function[{x, y, z, θ}, ColorData["TemperatureMap"][Sin[θ]^4]],
  ColorFunctionScaling → False, Mesh → True, AxesLabel → {x, y, z}]
DensityPlot[x, {x, 0, 1}, {y, 0, 0.05}, ColorFunction → ColorData["TemperatureMap"],
  ColorFunctionScaling → False, AspectRatio → Automatic,
  PlotRangePadding → None, FrameTicks → {{None, None}, {Automatic, None}},
  PlotLabel → "Potential vs Color in Units of Vo"]
```



Because the boundary is a sphere, it is convenient to express the solution in spherical coordinates. Further, the potential at the boundary does not depend on  $\phi$ , so the resulting potential will not depend on  $\phi$ . In this sense, the solution is “two-dimensional” (depending on  $r$  and  $\theta$  only), even though the potential and electric field are defined at every point in 3D space.

Laplace’s equation in spherical coordinates, with no  $\phi$  dependence, is given by:

$$\frac{1}{r^2} \frac{\partial}{\partial r} \left( r^2 \frac{\partial V[r, \theta]}{\partial r} \right) + \frac{1}{r^2 \sin \theta} \frac{\partial}{\partial \theta} \left( \sin \theta \frac{\partial V[r, \theta]}{\partial \theta} \right) = 0$$

One can generate solutions for this partial differential equation by separating the variables. One starts by assuming a solution of the form  $V[r, \theta] = R[r] \Theta[\theta]$ . With some algebra, one can often reduce the result to two ordinary differential equations (easier!), one in  $R[r]$  and one in  $\Theta[\theta]$ .

(a) Use M’s Laplacian command to calculate  $\nabla^2[R[r] \Theta[\theta]]$ . The syntax of the command is:

```
Laplacian[V[r,θ], {r,θ,φ}, "Spherical"]
```

The Expand option produces a simpler result in this case. You can also name the solution for future reference, for instance:

```
Lap1 = Laplacian[V[r,θ], {r,θ,φ}, "Spherical"]//Expand
```

```
In[9]:= (* Input code here *)
```

```
ClearAll["Global`*"]
```

```
Lap1 = Laplacian[R[r] Th[θ], {r, θ, φ}, "Spherical"] // Expand
```

```
Out[10]= 
$$\frac{2 \text{Th}[\theta] R'[r]}{r} + \frac{\text{Cot}[\theta] R[r] \text{Th}'[\theta]}{r^2} + \text{Th}[\theta] R''[r] + \frac{R[r] \text{Th}''[\theta]}{r^2}$$

```

(b) Multiplying each term by  $r^2$  and dividing by  $R[r] \Theta[\theta]$  will eliminate the  $r$ -dependence in the terms in  $\text{Th}'[\theta]$  and  $\text{Th}''[\theta]$  and eliminate the  $\theta$ -dependence from the terms in  $R'[r]$  and  $R[r]''$ . Use M to perform this multiplication. The simplify option helps. If you don’t want to copy the entire solution, you can refer to it by the name we gave it above.

In[11]:= (\* Input code here \*)

Lap2 = Lap1 \*  $\left(\frac{r^2}{R[r] \text{Th}[\theta]}\right)$  // Simplify

Out[11]=  $\frac{r (2 R'[r] + r R''[r])}{R[r]} + \frac{\text{Cot}[\theta] \text{Th}'[\theta] + \text{Th}''[\theta]}{\text{Th}[\theta]}$

If all went well, one set of terms will depend on  $r$  only, and the other set of terms will depend on  $\theta$  only. Since  $r$  and  $\theta$  are independent variables, it is not likely that any function of  $r$  will sum to zero with a function of  $\theta$  unless both functions are constant. As you will see, it is useful to choose a constant of the form  $l(l+1)$ , where  $l$  is a non-negative integer:  $l = 0, 1, 2, 3, \dots$  (In text files, I use an italics  $l$  from the Consolas font to avoid misreading it as a 1.)

We can define an ordinary differential equation for  $R[r]$  by selecting its terms and setting them equal to  $l(l+1)$ . To ensure a zero sum, we can define a second differential equation with  $\text{Th}[\theta]$  and its terms and set it equal to  $-l(l+1)$ . This choice of “separation constants” will ensure that Laplace’s equation is satisfied by any solution of the form  $R[r] \Theta[\theta]$  for a given value of  $l$ .

In[12]:= (\* Leave this code for students to execute \*)

Clear[r];

rOnly =  $\frac{r (2 R'[r] + r R''[r])}{R[r]} == l (l + 1)$

thOnly =  $\frac{\text{Cot}[\theta] \text{Th}'[\theta] + \text{Th}''[\theta]}{\text{Th}[\theta]} == -l (l + 1)$

Out[13]=  $\frac{r (2 R'[r] + r R''[r])}{R[r]} == l (l + 1)$

Out[14]=  $\frac{\text{Cot}[\theta] \text{Th}'[\theta] + \text{Th}''[\theta]}{\text{Th}[\theta]} == -l (l + 1)$

**(c) Use DSolve to solve these differential equations. Extract and simplify the solutions as necessary.**

In[15]:= (\* Input code below \*)

Rsol = DSolve[rOnly, R[r], r] // Simplify

Thsol = DSolve[thOnly, Th[θ], θ] // Simplify

Out[15]=  $\left\{ \left\{ R[r] \rightarrow r^{-\frac{1}{2}-\frac{1}{2}i\sqrt{l}\sqrt{l+1}} \sqrt{-4-\frac{1}{l+1^2}} \left( C[1] + r^{i\sqrt{l}\sqrt{l+1}} \sqrt{-4-\frac{1}{l+1^2}} C[2] \right) \right\} \right\}$

Out[16]=  $\left\{ \left\{ \text{Th}[\theta] \rightarrow C[1] \text{LegendreP}[l, \text{Cos}[\theta]] + C[2] \text{LegendreQ}[l, \text{Cos}[\theta]] \right\} \right\}$

**As usual, DSolve lists the solutions in brackets. Solution functions can be extracted from the list using the replacement command. For instance, R[r\_] =**

$R[r]/.$ Rsol[[1]]. The double brackets enclosing the “1” in Rsol[[1]] tell M to look for the first solution to the equation Rsol that is enclosed in two brackets. The resulting  $R[r]$  can be used to define a function  $R[r\_]$ , although it is often safer to rename the function. (M overwrites  $R[r]$ .)

```
In[17]:= R[r_] = R[r] /. Rsol[[1]]
Th[θ_] = Th[θ] /. Thsol[[1]]
```

$$\text{Out[17]= } r^{-\frac{1}{2}-\frac{1}{2}i\sqrt{l}\sqrt{1+l}} \sqrt{-4-\frac{1}{l+1^2}} \left( C[1] + r^{i\sqrt{l}\sqrt{1+l}} \sqrt{-4-\frac{1}{l+1^2}} C[2] \right)$$

```
Out[18]= C[1] LegendreP[l, Cos[θ]] + C[2] LegendreQ[l, Cos[θ]]
```

If your solution for  $R[r]$  looks like mine, it is complex. M assumes  $l$  is complex, and that makes  $R[r]$  complex. With pen and paper, you can show that the solution for  $R[r]$  can be written as:  $A[l]r^l + B[l]r^{-1-2l}$ , provided that  $l$  is real and not negative. The Simplify command can show this, but not as a “PostFix” operation—that is, not using the //Simplify notation. As a PostFix, Simplify does not recognize assumptions. Assumptions can be added as options to the full-blown Simplify command. For instance:

**Simplify[R[r], Assumptions→{l ∈ Integers && l ≥ 0}]**

**Try it! (O.K., I’ll do it for you):**

```
In[19]:= (* Input code for Simplify operation here *)
```

```
Simplify[R[r], Assumptions → {l ∈ Integers && l ≥ 0}]
```

```
Out[19]= r^l (C[1] + r^{-1-2l} C[2])
```

(d) The above solution can be rewritten as  $A[l]r^l + B[l]r^{-1-2l}$ . Which solutions ( $\propto r^l$  or  $\propto r^{-1-2l}$ ) are appropriate for potentials inside the sphere? Which are appropriate for solutions outside the sphere? Do these equations make any assumptions about the location of the zero potential?

<Enter answers below>

Solutions of the form  $r^l$  are appropriate for solutions inside a spherical boundary, because they converge as  $r \rightarrow 0$  (the center of the sphere). Conversely, solutions of the form  $r^{-(l+1)}$  are appropriate for solutions outside a spherical boundary, because they converge as  $r \rightarrow \infty$ . [Since  $l$  is a non-negative integer, the exponent  $(l+1)$  in  $r^{-(l+1)}$  never equals zero. Therefore all of these terms go to zero as  $r \rightarrow \infty$ .]

The potential for  $V_{\text{out}}[r,\theta]$  goes to zero as  $r \rightarrow \infty$ , no matter how the constant  $C[2]$  is chosen.

Like the Coulomb potential, the zero of potential in the series expansion is found as  $r \rightarrow \infty$ .

(e) You should have obtained two solutions for the angle dependence,  $TTh[\theta]$ : Legendre polynomials of the first kind,  $LegendreP[l, Cos[\theta]]$ , and Legendre polynomials of the second kind,  $LegendreQ[l, Cos[\theta]]$ .



Plot the first five LegendreP and the first five LegendreQ polynomials (for  $l = 0, 1, 2, 3,$  and  $4$ ) over the range  $\{\theta, 0, \pi\}$  and answer the following questions:

Which of these functions are suitable solutions for the  $\theta$ -dependence of the potential in spherical coordinates? Why?

In[20]:= (\* Input code here \*)

```
ClearAll["Global`*"] (* Leave ClearAll statement *)
Plot[{LegendreP[0, Cos[θ]], LegendreP[1, Cos[θ]],
      LegendreP[2, Cos[θ]], LegendreP[3, Cos[θ]], LegendreP[4, Cos[θ]]},
      {θ, 0, π}, PlotLabel → "First Five Legendre Polynomials of the First Kind",
      PlotLegends → "Expressions"]
```

```
ClearAll["Global`*"]
Plot[{LegendreQ[0, Cos[θ]], LegendreQ[1, Cos[θ]], LegendreQ[2, Cos[θ]],
      LegendreQ[3, Cos[θ]], LegendreQ[4, Cos[θ]]}, {θ, 0, π},
      PlotLabel → "First Five Legendre Polynomials of the Second Kind",
      PlotLegends → "Expressions"]
```



<Answer questions in this text cell>

The potential should be continuous and finite over the range of possible  $\theta$ -values:  $\{\theta, 0, \pi\}$ .

The Legendre polynomials of the first kind,  $\text{LegendreP}[L, \theta]$ , appear to be well-behaved (finite-valued and continuous) over the entire range. They should work.

The Legendre polynomials of the second kind,  $\text{LegendreQ}[l, \theta]$ , diverge as  $\theta \rightarrow 0$  and as  $\theta \rightarrow \pi$ . They are therefore unsuitable as solutions for the angular dependence of the potential over the range  $\{0, \pi\}$ . They might be of use in a problem where the  $\theta$ -dependence is somehow restricted. Not today.

(f) Write general, series solutions of Laplace's equation in spherical coordinates for the potentials inside and outside the sphere:  $V_{in}[r, \theta]$  and  $V_{out}[r, \theta]$ . Each series will be a linear combination of terms including Legendre polynomials,  $\text{LegendreP}[l, \text{Cos}[\theta]]$  with undetermined coefficients, say  $A[l]$  for  $V_{in}[r, \theta]$  and  $B[l]$  for  $V_{out}[r, \theta]$ .

In[24]:= (\* Input code here \*)

```
ClearAll["`*"] (* Leave ClearAll statement *)
Assumptions → l ∈ Integers && l ≥ 0 && {r, θ} ∈ Reals && r > 0 && 0 ≤ θ ≤ π;
(* Leave Assumptions *)
Vin[r_, θ_] = ∑l=0∞ A[l] rl LegendreP[l, Cos[θ]]
Vout[r_, θ_] = ∑l=0∞ B[l] r-(l+1) LegendreP[l, Cos[θ]]
```

Out[26]=  $\sum_{l=0}^{\infty} r^l A[l] \text{LegendreP}[l, \text{Cos}[\theta]]$

Out[27]=  $\sum_{l=0}^{\infty} r^{-l-1} B[l] \text{LegendreP}[l, \text{Cos}[\theta]]$

(g) Choose a few terms from each solution and show that they satisfy Laplace's equation in spherical coordinates. Show your results for  $V_{in}[r, \theta]$  and  $V_{out}[r, \theta]$  for one of these  $l$ -values. The Laplacian command with the //Simplify option is probably the easiest way to do this:

**Laplacian[your function here, {r, θ, φ}, "Spherical"]//Simplify**

In[28]:= (\* Input code here \*)

```
ClearAll["`*"] (* Leave ClearAll statement *)
l = 5;
Laplacian[A[l] rl LegendreP[l, Cos[θ]], {r, θ, φ}, "Spherical"] // Simplify
Laplacian[B[l] r-(l+1) LegendreP[l, Cos[θ]], {r, θ, φ}, "Spherical"] // Simplify
```

Out[30]= 0

Out[31]= 0

<Input any comments in this text cell>

M says that the Laplacian for the  $l = 5$  terms in each series is zero. I haven't tested all the possible values of  $l$ , but if each term (by itself) satisfies Laplace's equation, any sum of such terms will also satisfy Laplace's equation. We leave the general proof for another day.

(h) Write out the boundary conditions in terms of these expansions.

In[32]:= (\* Input code here \*)

$$V_{in}[R, \theta] == V_0 \sin[\theta]^4$$

$$V_{out}[R, \theta] == V_0 \sin[\theta]^4$$

Out[32]=  $V_{in}[R, \theta] == V_0 \sin[\theta]^4$

Out[33]=  $V_{out}[R, \theta] == V_0 \sin[\theta]^4$

The constants  $\{A[l]\}$  and  $\{B[l]\}$  may be determined from the boundary conditions by using the Fourier trick. You multiply both sides of the boundary condition equations by the  $m$ th order Legendre polynomial,  $\text{LegendreP}[m, \text{Cos}[\theta]] \times \text{Sin}[\theta]$  and integrate from 0 to  $\pi$ . (The factor of  $\text{Sin}[\theta]$  appears when you transform the argument of  $\text{LegendreP}[m, x]$  from  $x$  to  $\text{Cos}[\theta]$ .) You must not assume that the order  $m$  equals  $l$  in  $\text{LegendreP}[l, \text{Cos}[\theta]]$  above.

The integrals for the Fourier trick have the following form:

$$\begin{aligned} \int_0^\pi V_{in}[R, \theta] \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta &= \\ \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta &= \\ \int_0^\pi V_{out}[R, \theta] \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta &= \\ \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta &= \end{aligned}$$

You will also need the orthogonality relation for Legendre polynomials.

$$\int_0^\pi P[m, \text{Cos}[\theta]] P[n, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta = \begin{cases} 0 & m \neq n \\ \frac{2}{2l+1} & m = n \end{cases} = \frac{2 \delta_{mn}}{2l+1}$$

(i) Use your series expansions for  $V_{in}[]$  and  $V_{out}[]$  and the orthogonality relation to simplify the LHS of the "Fourier Trick" expression. It is safest to do this by hand.

<Input derivation in this text cell>

The left hand side of the Fourier trick expression is

$$\int_0^\pi V_{in}[R, \theta] \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta$$

where  $V_{in}[r, \theta] = \sum_{l=0}^{\infty} A[l] R^l \text{LegendreP}[l, \text{Cos}[\theta]]$  from Part (a) above. Replacing  $V_{in}[R, \theta]$  with the corresponding infinite series gives:

$$\int_0^{\pi} \sum_{l=0}^{\infty} A[l] R^l \text{LegendreP}[l, \text{Cos}[\theta]] \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta$$

The factors that do not depend on  $\theta$  can be taken outside the integral. This yields:

$$\sum_{l=0}^{\infty} A[l] R^l \int_0^{\pi} \text{LegendreP}[l, \text{Cos}[\theta]] \text{LegendreP}[m, \text{Cos}[\theta]] \text{Sin}[\theta] d\theta$$

The form of the expression inside the integral now matches the orthogonality relation, and therefore equals  $\frac{2 \delta_{lm}}{2l+1}$ . The LHS then becomes:

$$\sum_{l=0}^{\infty} \frac{2 A[l] R^l \delta_{lm}}{2l+1} = \frac{2 A[m] R^m}{2m+1}$$

The only nonzero term in the sum is that for which  $l = m$ , leaving

$$\text{LHS for } V_{in} = \frac{2 A[m] R^m}{2m+1}$$

The boundary condition for  $V_{out}[r, \theta]$  differs from the boundary condition for  $V_{in}[r, \theta]$  in two respects:  $A[l]$  is replaced by  $B[l]$ , and  $R^l$  is replaced by  $R^{-(l+1)}$ . Repeating the same series of operations on the boundary conditions for  $V_{out}$  yields.

$$\text{LHS for } V_{out} = \frac{2 B[m] R^{-(m+1)}}{2m+1}$$

(j) Now consider the RHS, which is the same for  $V_{in}[r, \theta]$  and  $V_{out}[r, \theta]$ . In general, the integral will depend on  $m$ , the order of the Legendre polynomial.

Use M to evaluate the RHS for  $m = 0$  through 10. Put the results in a table. Comment on your results. For instance, some terms are zero. If you can, explain why the zero terms might be expected to equal zero.

In[34]:= (\* Input code here \*)

ClearAll["`\*"] (\* Leave ClearAll Statement \*)

$$\text{RHS}[m_] = \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[m, \text{Cos}[\theta]] \sin[\theta] d\theta$$

RHSTable = Column[Table[{m, RHS[m]}, {m, 0, 10}]]

Out[35]=  $\int_0^\pi V_0 \text{LegendreP}[m, \text{Cos}[\theta]] \sin[\theta]^5 d\theta$

$$\{0, \frac{16 V_0}{15}\}$$

$$\{1, 0\}$$

$$\{2, -\frac{32 V_0}{105}\}$$

$$\{3, 0\}$$

Out[36]=  $\{4, \frac{16 V_0}{315}\}$

$$\{5, 0\}$$

$$\{6, 0\}$$

$$\{7, 0\}$$

$$\{8, 0\}$$

$$\{9, 0\}$$

$$\{10, 0\}$$

<Input comments in this text cell>

The RHS is zero except for  $m = 0, 2,$  and  $4$ . These are the first three “even” numbered Legendre polynomials. The even numbered Legendre polynomials include only even powers of  $\text{Cos}[\theta]$ .

$\text{Sin}[\theta]^4$  can be expressed as the sum of three terms with the form  $\text{Cos}[\theta]^n$ , where  $n$  is even.

$$\text{Sin}[\theta]^4 = (1 - \text{Cos}[\theta]^2)^2 = 1 - 2 \text{Cos}[\theta]^2 + \text{Cos}[\theta]^4$$

The same three powers of  $\text{Cos}[\theta]^n$  ( $n = 0, 2, 4$ ) appear in  $\text{LegendreP}[4, \text{Cos}[\theta]]$ . Note that the  $\text{LegendreP}[2, \text{Cos}[\theta]]$  polynomial includes the first two powers of  $\text{Cos}[\theta]^n$  ( $n = 0, 2$ ). Likewise the  $\text{LegendreP}[0, \text{Cos}[\theta]]$  polynomial includes only the  $n = 0$  term. So there ought to be a linear combination of  $\text{LegendreP}[0, \text{Cos}[\theta]]$ ,  $\text{LegendreP}[2, \text{Cos}[\theta]]$ , and  $\text{LegendreP}[4, -\text{Cos}[\theta]]$  that sums to  $\text{Sin}[\theta]^4$ . The Fourier trick is one way of determining how much weight to give each Legendre polynomial in the sum.

Since there are only three nonzero terms in the series, we may conclude that the sum of these three terms yields the “exact” solution to the potential. The result is not an approximation. Nice.

(k) Derive expressions for  $A[L]$  and  $B[L]$ . Put the results in a table. I left the integral from the RHS side in my expressions, as it is simple enough to recalculate the value of the integral. Another approach would be to copy your results for the

integration from Part (e).

<Input derivation in this text cell>

Equating LHS to RHS we have

$$\text{LHSin} = \frac{2R^l}{2^{l+1}} A[l] = \text{RHS} = \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$$

So that:

$$A[l] = \frac{2^{l+1}}{2R^l} \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$$

Similarly

LHSout =

$$\frac{2}{(2^{l+1}) R^{(l+1)}} B[l] = \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$$

Solving for B[l] gives:

$$B[l] = \frac{(2^{l+1}) R^{(l+1)}}{2} \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$$

We know that only the  $l = 0, 2,$  and  $4$  terms of the series are nonzero. I calculate a few more that yield zeros

The M-code I used follows.

In[37]:= (\* Input code here \*)

```
ClearAll["`*"]
Assumptions → {l, m} ∈ Integers && {l, m} ≥ 0; (* Leave Assumptions *)
lmax = 10;
A[l_] =  $\frac{2l+1}{2R^l} \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$ 
B[l_] =  $\frac{(2l+1)R^{(l+1)}}{2} \int_0^\pi V_0 \sin[\theta]^4 \text{LegendreP}[l, \cos[\theta]] \sin[\theta] d\theta$ 
Column[Table[{l, A[l], B[l]}, {l, 0, lmax}]]
```

$$\text{Out[40]= } \frac{1}{2} (1 + 2l) R^{-l} \int_0^\pi V_0 \text{LegendreP}[l, \cos[\theta]] \sin[\theta]^5 d\theta$$

$$\text{Out[41]= } \frac{1}{2} (1 + 2l) R^{1+l} \int_0^\pi V_0 \text{LegendreP}[l, \cos[\theta]] \sin[\theta]^5 d\theta$$

$$\begin{aligned} & \{0, \frac{8V_0}{15}, \frac{8R V_0}{15}\} \\ & \{1, 0, 0\} \\ & \{2, -\frac{16V_0}{21R^2}, -\frac{16R^3 V_0}{21}\} \\ & \{3, 0, 0\} \\ \text{Out[42]= } & \{4, \frac{8V_0}{35R^4}, \frac{8R^5 V_0}{35}\} \\ & \{5, 0, 0\} \\ & \{6, 0, 0\} \\ & \{7, 0, 0\} \\ & \{8, 0, 0\} \\ & \{9, 0, 0\} \\ & \{10, 0, 0\} \end{aligned}$$

**(l) Use your values of A[l] and B[l] to compute Vin[r,θ] and Vout[r,θ] in terms of Cos[θ] and r. If you do not clear your constants, M will remember and use the above values.**

In[43]:= (\* Input code here \*)

$$V_{in}[r_, \theta_] = \sum_{l=0}^{lmax} A[l] r^l \text{LegendreP}[l, \cos[\theta]]$$

$$V_{out}[r_, \theta_] = \sum_{l=0}^{lmax} B[l] r^{-(l+1)} \text{LegendreP}[l, \cos[\theta]]$$

$$\text{Out[43]= } \frac{8V_0}{15} - \frac{8r^2 V_0 (-1 + 3 \cos[\theta]^2)}{21R^2} + \frac{r^4 V_0 (3 - 30 \cos[\theta]^2 + 35 \cos[\theta]^4)}{35R^4}$$

$$\text{Out[44]= } \frac{8R V_0}{15r} - \frac{8R^3 V_0 (-1 + 3 \cos[\theta]^2)}{21r^3} + \frac{R^5 V_0 (3 - 30 \cos[\theta]^2 + 35 \cos[\theta]^4)}{35r^5}$$

**(m) Use Plot3D[] to plot the potential inside and outside the sphere for reasonable choices of R and Vo. I set R = 1 m and Vo = 1 V. Remember that Plot3D[] requires the function to be expressed in terms of Cartesian coordinates, {x,y,z}. You will**

have to choose two of the three Cartesian coordinates, as you need one coordinate axis to show the potential values.

Will you plot  $V[x,y]$ ,  $V[y,z]$ , or  $V[z,x]$ ? Why? What substitutions are needed to express the above potential in terms of your Cartesian coordinates? Write and run your M code.

<Describe your plan here>

I plotted  $V[y,z]$ , assuming that  $x = 0$ . One expects the potential to depend on  $x$  and  $y$  in similar ways because of the symmetry. By symmetry, rotations about the  $z$ -axis will not change the potential anywhere on the sphere. For instance, one can “rotate” the  $x$ -axis around the  $z$ -axis so that it lies on top of the old  $y$ -axis.

The  $z$ -axis is special.  $\text{Cos}[\theta] = z/r$ . With  $z$ , either  $x$  or  $y$  would be a good choice. I plotted  $V[y,z]$ . The following “replacements” were required:

$$r \rightarrow (x^2 + y^2)^{1/2}$$

$$\text{Cos}[\theta] \rightarrow z/r.$$

In[45]:= (\* Input code here \*)

```


Vo = 1; R = 1;
r = (y^2 + z^2)^(1/2);
Vin[y_, z_] =
  ( (8/15 - 8/21 r^2 (-1 + 3 Cos[θ]^2)) + 1/35 r^4 (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) ) /. Cos[θ] -> z / r
Vout[y_, z_] = ( (8/15 r - 8 (-1 + 3 Cos[θ]^2) / (21 r^3) + (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) / (35 r^5) ) /. Cos[θ] -> z / r
Plot3D[If[r < R, Vin[y, z], Vout[y, z]],
  {y, -2, 2}, {z, -2, 2}, AxesLabel -> {y, z, V[y, z]},
  PlotLabel -> "Potential Inside and Outside Sphere with V[θ] = Sin[θ]^4"]

```

Out[47]= 
$$\frac{8}{15} + \frac{1}{35} (y^2 + z^2)^2 \left( 3 + \frac{35 z^4}{(y^2 + z^2)^2} - \frac{30 z^2}{y^2 + z^2} \right) - \frac{8}{21} (y^2 + z^2) \left( -1 + \frac{3 z^2}{y^2 + z^2} \right)$$

Out[48]= 
$$\frac{8}{15 \sqrt{y^2 + z^2}} + \frac{3 + \frac{35 z^4}{(y^2 + z^2)^2} - \frac{30 z^2}{y^2 + z^2}}{35 (y^2 + z^2)^{5/2}} - \frac{8 \left( -1 + \frac{3 z^2}{y^2 + z^2} \right)}{21 (y^2 + z^2)^{3/2}}$$

Potential Inside and Outside Sphere with V[θ] = Sin[θ]^4



(n) Describe your plot below. Where is the z-axis of the sphere? Where is the equator of the sphere? Do the peaks and troughs of the potential function match the maximum and minimum values of the Sin[θ]^4 function?

<Input your description in this text cell>

I plotted V[x=0,y,z]: the potential as a function y and z for the special case of x = 0. The x-values are not shown. The z-axis is the conventional z-axis for spherical coordinates. The N and S poles of the sphere are located at points {y,z} = {0,±1}. Of the points lying on the equator, the plot shows only two, {y,z} = {±1,0}.

The maximum potential of  $V_0 = 1$  is reached at points  $\{y,z\} = \{\pm 1,0\}$ , which corresponds to  $\theta = \pi/2$ , on the equator of the sphere. This coincides the the maximum value of  $\text{Sin}[\theta]^4$  when  $\theta = \pi/2$ . Likewise, the minimum value of  $\text{Sin}[\theta]^4 = 0$  is reached at  $\theta = 0$  and  $\pi$ , which corresponds to points  $\{y,z\} = \{0,\pm 1\}$  at the poles of the sphere.

It is remarkable how quickly the  $\theta$ -dependence of the potential drops as you move away from the sphere. In the limit of large  $r$ , we expect the potential to vary as  $r^{-1}$ , like the potential of a point charge. The first term in the series for  $V_{out}[r,\theta]$  varies as  $r^{-1}$  does not depend on  $\theta$ . The higher order terms (which do depend on  $\theta$ ) both decay quickly with increasing  $r$  (as  $r^{-3}$  and  $r^{-5}$ ). These higher-order terms are responsible for the fast decay observed above.

(o) Before we plot the electric field, it is useful to show that the field component that you do *not* plot is zero. For instance, if you calculated  $V[y,z]$  above, you will want to plot the electric field in the y-z plane, where  $x = 0$ . If the x-component of the field is *not* zero when  $x = 0$ , your plot will not show it. That would be bad.

First, write  $V_{in}[r,\theta]$  and  $V_{out}[r,\theta]$  in terms of all three Cartesian components. Then use the `D[]` command (partial derivative) to calculate the one component of the electric field that you will not show in your field plot. For instance, if I hope to plot the field in the y-z plane, I hope that the x-component of the field in the y-z plane is zero. Use the `replace all` command (for instance, `/.x→0`) to see if the x-component of the field zero when  $x = 0$ .

In[50]:= (\* Input code here \*)

```

Vo = 1; R = 1;
r = (x^2 + y^2 + z^2)^(1/2);
Vin[x_, y_, z_] =
  ( (8/15 - 8/21 r^2 (-1 + 3 Cos[θ]^2) + 1/35 r^4 (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) ) /. Cos[θ] → z / r;
Vout[x_, y_, z_] = ( (8/15 r - 8 (-1 + 3 Cos[θ]^2) / (21 r^3) + (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) / (35 r^5) ) /.
  Cos[θ] → z / r;
-D[Vin[x, y, z], x] /. x → 0      (* x-component of Vin for x = 0 *)
-D[Vout[x, y, z], x] /. x → 0     (* x-component of Vout for x = 0 *)

```

Out[54]= 0

Out[55]= 0

<Describe your results in this text cell>

I plan to plot the field in the y-z plane, where x is zero. The x-component of the field is the

partial derivative of the potential with respect to  $x$ . When I set  $x$  to zero, both partial derivatives are zero. When I plot the fields as functions of  $y$  and  $z$  (with  $x$  set to zero), I can be confident that my plot shows the true direction of the electric field. When  $x = 0$ , the all the field vectors will lie *in* the  $y$ - $z$  plane.

(p) Construct a ContourPlot of the potential and a StreamPlot of the electric field. Superimpose them. Describe the result. Do you see evidence for negative surface charge on the sphere?

In[56]:= (\* Input code here \*)



```


Vo = 1; R = 1;
r = (y^2 + z^2)^(1/2);
Vin[y_, z_] =
  ( (8/15 - 8/21 r^2 (-1 + 3 Cos[θ]^2) + 1/35 r^4 (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) ) /. Cos[θ] → z / r;
Vout[y_, z_] = ( (8/15 r - 8 (-1 + 3 Cos[θ]^2) / (21 r^3) + (3 - 30 Cos[θ]^2 + 35 Cos[θ]^4) / (35 r^5) ) /. Cos[θ] → z / r;
EEin[y_, z_] = -Grad[Vin[y, z], {y, z}] // Simplify
EEout[y_, z_] = -Grad[Vout[y, z], {y, z}] // Simplify
contoury = ContourPlot[If[r < R, Vin[y, z], Vout[y, z]], {y, -2, 2},
  {z, -2, 2}, PlotLabel → "V[y,z] for Sin[θ^4] Potential Distribution"]
streamy = StreamPlot[If[r < R, EEin[y, z], EEout[y, z]], {y, -2, 2}, {z, -2, 2}]
Show[contoury, streamy]

```

Out[60]=  $\left\{ -\frac{4}{105} y (20 + 9 y^2 - 36 z^2), \frac{16}{105} z (10 + 9 y^2 - 6 z^2) \right\}$

Out[61]=  $\left\{ (56 y^9 + 8 y^7 (15 + 28 z^2) + 8 y z^4 (45 - 60 z^2 + 7 z^4) + 4 y^3 z^2 (-135 - 210 z^2 + 56 z^4) + 3 y^5 (15 - 80 z^2 + 112 z^4)) / (105 (y^2 + z^2)^{11/2}), \right.$   
 $\left. (z (56 y^8 + 8 y^6 (45 + 28 z^2) + 8 z^4 (15 - 30 z^2 + 7 z^4) + 8 y^2 z^2 (-75 - 15 z^2 + 28 z^4) + 3 y^4 (75 + 160 z^2 + 112 z^4)) / (105 (y^2 + z^2)^{11/2}) \right\}$





<Input your description in this text file>

As expected, the electric field is directed outward from parts of the sphere with high positive potentials, near the equator. Interestingly, the electric field is directed inward toward parts of the sphere with low (but still positive) potentials, near the poles of the sphere.

The presence of field lines directed inward, toward the surface, at the poles of the sphere indicates the presence of negative surface charge in these regions. The absence of negative potential does not imply the absence of negative charge.

The D-shaped contours at the poles of the sphere are interesting. I don't know what to make of them, but they are interesting.

(q) Griffiths claims that the potential at the center of a sphere is equal to the average potential on the surface of the sphere (Section 3.1.4).

Find the average potential of the surface of the sphere and compare it to the potential at the center of the sphere.

The average potential on the surface of the sphere is given by

$$\int V[R, \theta, \phi] dA \div \int dA$$

where  $dA$  is the surface area element,  $R^2 \sin[\theta] d\theta d\phi = 2\pi R^2 \sin[\theta] d\theta$ . The surface area of

the sphere is given by  $\int dA = 4\pi R^2$ . For  $V_0 = 1$ , the average is given by:

In[65]:= (\* Input code for average potential on sphere surface here \*)

Average = Integrate[Sin[ $\theta$ ]<sup>4</sup> 2  $\pi$  Sin[ $\theta$ ], { $\theta$ , 0,  $\pi$ }] / (4  $\pi$ )

Out[65]=  $\frac{8}{15}$

The potential at the center of the sphere is given by the first term in the Legendre expansion for  $V_{in}[r,\theta]$ , that is  $A[0]$ . (All the other terms in the Legendre series for  $V_{in}[r,\theta]$  equal zero when  $r = 0$ .) If the variable  $A[0]$  has not been cleared or overwritten, you can still query  $M$  for its value. Otherwise, you can manually check your results above.

(\* Input code to find potential at center of sphere here \*)

$A[0]$

For  $R = 1$  and  $V_0 = 1$ , the average potential on the surface of the sphere is  $8/15$ . Under the same set of assumptions, the potential at the center of the sphere ( $=A[0]$ ) also equals  $8/15$ .

These results are consistent with Griffiths' theorem stating that the potential at the center of a sphere is equal to the average potential on the surface of the sphere (Section 3.1.4).

If you had never done the calculation, this theorem would tell you that the potential at the center of the sphere must *not* be zero, because the potential on the surface of the sphere is zero or positive everywhere (so its average over that surface is not zero).

Note that the electric field is a different beast: it is zero at the center of the sphere (easily seen by symmetry).